Zephir Documentation

Release 0.3.0

Zephir Team

Contents

Kerei	rence
1.1	Welcome!
1.2	Why Zephir?
1.3	Introducing Zephir
1.4	Installation
1.5	Tutorial
1.6	Basic Syntax
1.7	Types
1.8	Operators
1.9	Arrays
1.10	Classes and Objects
1.11	Control Structures
1.12	Calling Functions
1.13	Using optimizers
1.14	License
Othe	r Formats
dex	
	1.1 1.2 1.3 1.4 1.5 1.6 1.7 1.8 1.9 1.10 1.11 1.12 1.13

Reference

1.1 Welcome!

Welcome to Zephir, an open source, high-level/domain specific language designed to ease the creation and maintainability of extensions for PHP with a focus on type and memory safety.

1.1.1 Some features

Zephir's main features are:

Type system	dynamic/static
Memory safety	pointers or direct memory management aren't allowed
Compilation model	ahead of time
Memory model	task-local garbage collection

1.1.2 A small taste

The following code registers a class with a method that filters variables returning its alphabetic characters:

```
namespace MyLibrary;

/**
    * Filter
    */
class Filter
{
        /**
          * Filters a string returning its alpha characters
          *
          * @param string str
          */
        public function alpha(string str)
        {
              char ch; string filtered = "";

              for ch in str {
                if (ch >= 'a' && ch <= 'z') || (ch >= 'A' && ch <= 'Z') {</pre>
```

```
let filtered .= ch;
}

return filtered;
}
```

The class can be used from PHP as follows:

```
<?php

$filter = new MyLibrary\Filter();
echo $filter->alpha("01he$1.lo?/1"); // prints hello
```

1.2 Why Zephir?

Today's PHP applications must balance a number of concerns including stability, performance and functionality. Every PHP application is based on a set of common components that are also base for most of the application.

These common components are libraries/frameworks or a combination of them. Once installed, frameworks rarely change, being the foundation of the application they must be highly functional and also very fast.

Getting fast and robust libraries can be complicated due to high levels of abstraction that are typically implemented on them. Given the condition that base libraries or frameworks rarely change, there is an opportunity to build extensions that provide this functionality taking advantage of the compilation improving performance and resource consumption.

With Zephir, you can implement object-oriented libraries/frameworks/applications that can be used from PHP gaining important seconds that can make your application faster while improving the user experience.

1.2.1 If You Are a PHP Programmer...

PHP is one of the most popular languages in use for the development of web applications. Dynamically typed and interpreted languages like PHP offer very high productivity due to their flexibility.

Since version 4 and then 5, PHP is based on the Zend Engine implementation. This is a virtual machine that executes the PHP code from its bytecode representation. Zend Engine is almost present in every PHP installation in the world, with Zephir, you can create extensions for PHP running under the Zend Engine.

PHP is hosting Zephir, so they obviously have a lot of similarities, however; they have important differences that give Zephir its own personality. For example, Zephir is more strict, and it could be make you less productive compared to PHP due to the compilation step.

1.2.2 If You Are a C Programmer...

C is one of the most powerful and popular languages ever created. In fact, PHP is written in C, which is one of the reasons why PHP extensions are available for it. C gives you the freedom to manage memory, use low level types and even inline assembly routines.

However, developing big applications in C can take much longer than expected compared to PHP or Zephir and some errors can be tricky to find if you aren't an experienced developer.

Zephir was designed to be safe, so it doesn't implement pointers or manual memory management, so if you're a C programmer, you will feel Zephir less powerful but more friendly than C.

1.2.3 Compilation vs Interpretation

Compilation usually slows the development down; you will need a bit more of patience to make your code compiled before running it. Moreover, the interpretation tends to reduce the performance in favor of productivity. In some cases, there is no any noticeable difference between the speed of the interpreted and compiled code.

Zephir requires compilation of your code, however, the functionality is used from PHP that is interpreted.

Once the code is compiled is not necessary to do so, however, interpreted code is interpreted each time it is run. A developer can decide which parts of your application should be in Zephir and which not.

1.2.4 Statically Typed Versus Dynamically Typed Languages

In general speaking, in a static typed language, a variable is bound to a particular type for its lifetime. Its type can't be changed and it can only reference type-compatible instances and operations. Languages like C/C++ were implemented with the scheme:

```
int a = 0;
a = "hello"; // not allowed
```

In dynamic typing, the type is bound to the value, not the variable. So, a variable might refer to a value of a type, then be reassigned later to a value of an unrelated type. Javascript/PHP are examples of a dynamic typed language:

```
var a = 0;
a = "hello"; // allowed
```

Despite their productivity advantages, dynamic languages may not be the best choices for all applications, particularly for very large code bases and high-performance applications.

Optimizing the performance of a dynamic language like PHP is more challenging than for a static language like C. In a static language, optimizers can exploit the type information to make decisions. In a dynamic language, fewer such clues are available for the optimizer, making optimization choices harder.

While recent advancements in optimizations for dynamic languages are promising (like JIT compilation), they lag behind the state of the art for static languages. So, if you require very high performance, static languages are probably a safer choice.

Another small benefit of static languages is the extra checking the compiler performs. A compiler can't find logic errors, which are far more significant, but a compiler can find errors in advance that in a dynamic language only can be found in runtime.

Zephir is both statically and dynamically typed allowing you to take advantage from both sides where possible.

1.2.5 Compilation Scheme

Zephir offers native code generation (currently via compilation to C), a compiler like gcc/clang/vc++ optimizes and compiles the code down to machine code. The following graph shows how the process works:

In addition to the ones provided by Zephir, Over time, compilers have been implemented and matured a number of optimizations that improve the performance of compiled applications:

- GCC optimizations
- LLVM passes
- Visual C/C++ optimizations

1.2. Why Zephir?

Zephir: Optimization + code generation gcc/clang/vc++: Compilation + Optimization Native execution

1.2.6 Code Protection

In some circumstances, the compilation does not significantly improve performance, this may be because the bottleneck is located in the I/O bound (quite likely) of the application rather than compute/memory bound. However, compiling code could also bring some level of intelectual protection to your application. With Zephir, producing binaries you also get the ability to hide the code behind users or customers.

1.2.7 Conclusion

Zephir was not created to replace PHP or C, instead of this, we think it is a complement to them, allowing developers to venture into code compilation and static typing. Zephir is precisely an attempt to join good things from the C and PHP worlds, looking for opportunities to make their applications faster.

1.3 Introducing Zephir

Zephir is a language that addresses the major needs of a PHP developer trying to write and compile code that can be executed by PHP. It is a dynamically/statically typed, some of its features can be familiar to PHP developers.

The name Zephir is a contraction of the words Zend Engine/PHP/Intermediate. While this suggests that the pronunciation should be zephyr, the creators of Zephir actually pronounce it zaefire.

1.3.1 Hello World!

Every language has its own "Hello World!" sample, in Zephir this introductory example showcases some important features of this language.

Code in Zephir must be placed in classes. This language is intended to create object-oriented libraries/frameworks, so code out of a class is not allowed. Also, a namespace is required:

```
* This is a sample method
*/
public function say()
{
    echo "Hello World!";
}
```

Once this class is compiled it produce the following code that is transparently compiled by gcc/clang/vc++:

```
#ifdef HAVE_CONFIG_H
#include "config.h"
#endif

#include "php.h"
#include "php_test.h"
#include "test.h"

#include "kernel/main.h"

/**
   * This is a sample class
   */
ZEPHIR_INIT_CLASS(Test_Hello) {
    ZEPHIR_REGISTER_CLASS(Test, Hello, hello, test_hello_method_entry, 0);
    return SUCCESS;
}

/**
   * This is a sample method
   */
PHP_METHOD(Test_Hello, say) {
    php_printf("%s", "Hello World!");
}
```

Actually, it is not expected that a developer that uses Zephir must know or even understand C, however, if you have any experience with compilers, php internals or the C language itself, it would provide a more clear sceneario to the developer when working with Zephir.

1.3.2 A Taste of Zephir

In the following examples, we'll describe just enough of the details, so you understand what's going on. The goal is to give you a sense of what programming in Zephir is like. We'll explore the details of the features in subsequent chapters.

The following example is very simple, it implements a class and a method with a small program that checks the types of an array

Let's examine the code in detail, so we can begin to learn Zephir syntax. There are a lot of details in just a few lines of code! We'll explain the general ideas here:

```
namespace Test;

/**
  * MyTest (test/mytest.zep)
  */
class MyTest
{
```

```
public function someMethod()
{
    /* Variables must be declared */
    var myArray;
    int i = 0, length;

    /* Create an array */
    let myArray = ["hello", 0, 100.25, false, null];

    /* Count the array into a 'int' variable */
    let length = count(myArray);

    /* Print value types */
    while i < length {
        echo typeof myArray[i], "\n";
        let i++;
    }

    return myArray;
}</pre>
```

In the method, the first lines use the 'var' and 'int' keywords are used to declare a variable in the local scope. Every variable used in a method must be declared with its respective type. This declaration is not optional, it helps the compiler to report you about mistyped variables or about the use of variables out of their scope which usually ends in runtime errors.

Dynamic variables are declared with the keyword 'var'. These variables can be assigned and reassigned to different types. On the other hand, we have 'i' and 'length' integer static typed variables that can only have values of this type in the entire program execution.

In contrast with PHP, you are not required to put a dollar sign (\$) in front of variable names.

Zephir follows the same comment conventions as Java, C#, C++, etc. A //comment goes to the end of a line, while a /* comment */ can cross line boundaries.

Variables are by default immutable, this means that Zephir expects that most variables stay unchanged. Variables that maintain their initial value can be optimized down by the compiler to static constants. When the variable value needs to be changed, the keyword 'let' must be used:

```
/* Create an array */
let myArray = ["hello", 0, 100.25, false, null];
```

By default, arrays are dynamic like in PHP, they may contain values of different types. Functions from the PHP userland can be called in Zephir code, in the example the function 'count' was called, the compiler can performs optimizations like avoid this call because it already knows the size of the array:

```
/* Count the array into a 'int' variable */
let length = count(myArray);
```

Parentheses in control flow statements are optional, you can also use them if you feel more confortable.

```
while i < length {
    echo typeof myArray[i], "\n";
    let i++;
}</pre>
```

PHP only works with dynamic variables, methods always return dynamic variables, this means that if a static typed variable is returned, in the PHP side, you will get a dynamic variable that can be used in PHP code. Note that memory

is automatically managed by the compiler, so you don't need to allocate or free memory like in C, working in a similar way than PHP.

1.4 Installation

To install Zephir, please follow these steps:

1.4.1 Prerequisites

To build a PHP extension and use Zephir you need the following requirements:

- gcc >= 4.x/clang >= 3.x/vc++ 9
- re2c 0.13 or later
- gnu make 3.81 or later
- autoconf 2.31 or later
- automake 1.14 or later
- php development headers and tools

Since Zephir is written in PHP you need to have installed a recent version of PHP and it must be available in your console:

```
$ php -v
PHP 5.5.7 (cli) (built: Dec 14 2013 00:44:43)
Copyright (c) 1997-2013 The PHP Group
Zend Engine v2.5.0, Copyright (c) 1998-2013 Zend Technologies
```

Also, make sure you have also the PHP development libraries installed along with your PHP installation:

```
$ phpize -v
Configuring for:
PHP Api Version: 20121113
Zend Module Api No: 20121212
Zend Extension Api No: 220121212
```

You don't have to necessarely see the exact above output but it's important that these commands are available to start developing with Zephir.

1.4.2 Installing Zephir

Json-C is required by the Zephir parser to process the code, you can install it this way:

```
$ git clone https://github.com/json-c/json-c.git
$ cd json-c
$ sh autogen.sh
$ ./configure
$ make
$ make install
```

The Zephir compiler currently must be cloned from Github:

```
$ git clone https://github.com/phalcon/zephir
```

1.4. Installation 7

Run the Zephir installer (this compiles/creates the parser):

```
$ cd zephir
$ ./install
```

Add the "bin" directory to your PATH environment variable:

```
$ export PATH=$PATH:/path/to/zephir/bin
```

1.4.3 Testing Installation

Check if Zephir is available from any directory by executing:

```
$ zephir help
```

1.5 Tutorial

Zephir and this book are intended for PHP developers which want to create C-extensions with a lower complexity.


We assume that you are experienced in one or more other programming languages. We draw parallels to features in PHP, C, Javascript, and other languages. If you know any of these languages, we'll point out similar features in Zephir, as well as many features that are new or different.

1.5.1 Checking Installation

If you have successfully installed Zephir, you must be able to execute the following command in your console:

```
$ zephir help
```

If everything is well, you should see the following help in your screen:



```
Zephir version 0.3.0a
```

Usage:

```
command [options]
```

Available commands:

```
build Generate/Compile/Install a Zephir extension clean Cleans the generated object files in compilation compile Compile a Zephir extension full-clean Cleans the generated object files in compilation generate Generates C code from the Zephir code help Displays this help init [namespace] Initializes a Zephir extension install Installs the extension (requires root password) version Shows Zephir version
```

1.5.2 Extension Skeleton

The first thing we have to do is generate an extension skeleton, this will provide to our extension the basic structure we need to start working. In our case, we're going to create an extension called "utils":

```
$ zephir init utils
```

After this, a directory called "utils" is created on the current working directory:

```
utils/
ext/
utils/
```

The directory "ext/" (inside utils) contains the code that is going to be used by the compiler to produce the extension. The another directory created is "utils", this directory has the same as our extension. We will place Zephir code in this directory.

We need to change the working directory to "utils" to start compiling our code:

```
$ cd utils
$ ls
ext/ utils/ config.json
```

The directory listing will also show us a file called "config.json", this file contains configuration settings we can use to alter the behavior of Zephir and/or this extension.

1.5.3 Adding our first class

Zephir is designed to generate object-oriented extensions. To start developing functionality we need to add our first class to the extension.

As in many languages/tools, the first thing we want to do is see a "hello world" generated by Zephir and check that everything is well. So our first class will be called "Utils\Greeting" and it contains a method printing "hello world!".

The code for this class must be placed in "utils/utils/greeting.zep":

```
namespace Utils;
class Greeting
{
    public static function say()
    {
        echo "hello world!";
    }
}
```

Now, we need to tell Zephir that our project must be compiled and the extension generated:

```
$ zephir build
```

Initially, and only for the first time, a number of internal commands are executed producing the necessary code and configurations to export this class to the PHP extension, if everything goes well you will see the following message at the end of the output:

```
Extension installed!
Add extension=utils.so to your php.ini
Don't forget to restart your web server
```

1.5. Tutorial 9

At the above step, it's likely that you would need to supply your root password in order to install the extension. Finally, the extension must be added to the php.ini in order to be loaded by PHP. This is achieved by adding the initialization directive: extension=utils.so to it.

1.5.4 Initial Testing

Now that the extension was added to your php.ini, check whether the extension is being loaded properly by executing the following:

```
$ php -m
[PHP Modules]
Core
date
libxml
pcre
Reflection
session
SPL
standard
tokenizer
utils
xdebug
xml
```

Extension "utils" must be part of the output indicating that the extension was loaded correctly. Now, let's see our "hello world" directly executed by PHP. To accomplish this, you can create a simple PHP file calling the static method we have just created:

```
<?php
echo Utils\Greeting::say(), "\n";</pre>
```

Congratulations!, you have your first extension running on PHP.

1.5.5 A useful class

The "hello world" class was fine to check if our environment was right, now, let's create some more useful classes.

The first useful class we are going to add to this extension will provide filtering facilities to users. This class is called "Utils\Filter" and its code must be placed in "utils/utils/filter.zep":

A basic skeleton to this class is the following:

```
namespace Utils;
class Filter
{
}
```

The class contains filtering methods that help users to filter unwanted caracters from strings. The first method is called "alpha" and its purpose is to filter only those characters that are ascii basic letters. To begin, we are just going to traverse the string printing every byte to the standard output:

```
namespace Utils;
class Filter
```

```
{
    public function alpha(string str)
         char ch;
         for ch in str {
             echo ch, "\n";
    }
}
When invoking this method:
<?php
$f = new Utils\Filter();
$f->alpha("hello");
You will see:
e
1
1
Checking every character in the string is straightforward, we now just could create another string with the right filtered
characters:
class Filter
    public function alpha(string str) -> string
         char ch; string filtered = "";
```

if (ch >= 'a' && ch <= 'z') || (ch >= 'A' && ch <= 'Z') {

The complete method can be tested as before:

for ch in str {

return filtered;

}

```
<?php

$f = new Utils\Filter();
echo $f->alpha("$he$0213'12110."); // prints "hello"
```

let filtered .= ch;

In the following screencast you can watch how to create the extension explained in this tutorial:

1.5. Tutorial

1.5.6 Conclusion

This is a very simple tutorial and as you can see, it's easy to start building extensions using Zephir. We invite you to continue reading the manual so that you can discover additional features offered by Zephir!

1.6 Basic Syntax

In this chapter, we'll discuss the organization of files and namespaces, variable declarations, miscellaneous syntax conventions, and a few other concepts.

1.6.1 Organizing Code in Files and Namespaces

In PHP, you can place the code in any file without a specific structure. In Zephir, every file must contain a class (and just one class). Every class must have a namespace and the directory structure must match the names of classes and namespaces used.

For example, given the following structure, the classes in each file must be:

Class in mylibrary/router.zep:

```
namespace MyLibrary;
class Router
{
}
```

Class in mylibrary/router/exception.zep:

```
namespace MyLibrary\Router;
class Router extends Exception
{
```

Zephir will raise a compiler exception if a file or class is not located at the expected file or vice versa.

1.6.2 Instruction separation

You may have already noticed that there were very few semicolons in the code examples in the previous chapter. You can use semicolons to separate statements and expressions, as in Java, C/C++, PHP, and similar languages:

```
myObject->myMethod(1, 2, 3); echo "world";
```

1.6.3 Comments

Zephir supports 'C'/'C++' comments, these are one line comments with // and multi line comments with /* ... */:

```
// this is one line comment
/**
  * multi-line comment
  */
```

In most languages, comments are simply text ignored by the compiler/interpreter. In Zephir, multi-line comments are also used as docblocks, and they're exported to the generated code, so they're part of the language!.

If a docblock is not located where is expected, the compiler will throw an exception.

1.6.4 Variable Declarations

In Zephir, all variables used in a given scope must be declared. This process gives important information to the compiler to perform optimizations and validations. Variables must be unique identifiers, and they cannot be reserved words.

```
//Declaring variables for the same type in the same instruction
var a, b c;

//Declaring each variable in different lines
var a;
var b;
var c;
```

Variables can optionally have an initial compatible default value, you can assign a new value to a variable as often as you want:

```
//Declaring variables with default values
var a = "hello", b = 0, c = 1.0;
int d = 50; bool some = true;
```

Variable names are case-sensitive, the following variables are different:

```
//Different variables
var somevalue, someValue, SomeValue;
```

1.6.5 Variable Scope

All variables declared are locally scoped to the method where they were declared:

```
namespace Test;

class MyClass
{

   public function someMethod1()
   {
      int a = 1, b = 2;
      return a + b;
   }

   public function someMethod2()
   {
      int a = 3, b = 4;
      return a + b;
   }
}
```

1.6. Basic Syntax 13

}

1.6.6 Super Globals

Zephir doesn't support global variables, accessing global variables from the PHP userland is not allowed. However, you can access the PHP's super-globals as follows:

```
//Getting a value from _POST
let price = _POST["price"];

//Read a value from _SERVER
let requestMethod = _SERVER["REQUEST_METHOD"];
```

1.6.7 Local Symbol Table

Every method or context in PHP has a symbol table that allows to write variables in a very dynamic way:

```
<?php
$b = 100;
$a = "b";
echo $$a; // prints 100</pre>
```

Zephir does not implement this feature since all variables are compiled down to low level variables and there is no way to know which variables do exist in a specific context. If you want to create a variable in the current PHP symbol table, you can use the following syntax:

```
//Set variable $name in PHP
let {"name"} = "hello";

//Set variable $price in PHP
let name = "price";
let {name} = 10.2;
```

1.7 Types

Zephir is both dynamic and static typed. In this chapter we highlight the supported types and its behavior:

1.7.1 Dynamic Type

Dynamic variables are exactly like the ones in PHP, they can be assigned and reassigned to different types without restriction.

A dynamic variable must be declared with the keyword 'var', the behavior is nearly the same as in PHP:

```
var a, b, c;

// Initialize variables
let a = "hello", b = false;

// Change their values
let a = 10, b = "140";
```

```
// Perform operations between them let c = a + b;
```

They can have eight types:

Type	Description
boolean	A boolean expresses a truth value. It can be either 'true' or 'false'.
integer	Integer numbers. The size of an integer is platform-dependent.
float/double	Floating point numbers. The size of a float is platform-dependent.
string	A string is series of characters, where a character is the same as a byte.
array	An array is an ordered map. A map is a type that associates values to keys
object	Object abstraction like in PHP
resource	A resource holds a reference to an external resource
null	The special NULL value represents a variable with no value

Check more info about these types in the PHP manual

Boolean

A boolean expresses a truth value. It can be either 'true' or 'false':

```
var a = false, b = true;
```

Integer

Integer numbers. The size of an integer is platform-dependent, although a maximum value of about two billion is the usual value (that's 32 bits signed). 64-bit platforms usually have a maximum value of about 9E18. PHP does not support unsigned integers so Zephir has this restriction too:

```
var a = 5, b = 10050;
```

Integer overflow

Contrary to PHP, Zephir does not automatically checks for integer overflows, like in C if you are doing operations that may return a big number you can use types such as 'unsigned long' or 'float' to store them:

```
unsigned long my_number = 2147483648;
```

Float/Double

Floating-point numbers (also known as "floats", "doubles", or "real numbers"). Floating-point literals are expressions with zero or more digits, followed by a period (.), followed by zero or more digits. The size of a float is platform-dependent, although a maximum of ~1.8e308 with a precision of roughly 14 decimal digits is a common value (the 64 bit IEEE format).

```
var number = 5.0, b = 0.014;
```

Floating point numbers have limited precision. Although it depends on the system, as PHP, Zephir uses the IEEE 754 double precision format, which will give a maximum relative error due to rounding in the order of 1.11e-16.

1.7. Types 15

String

A string is series of characters, where a character is the same as a byte. As PHP, Zephir only supports a 256-character set, and hence does not offer native Unicode support.

```
var today = "friday";
```

In Zephir, string literals can only be specified using double quotes (like in C), single quotes are reserved for chars.

The following escape sequences are supported in strings:

Sequence	Description
\t	Horizontal tab
\n	Line feed
\r	Carriage return
\\	Backslash
\"	double-quote

```
var today = "\tfriday\n\r",
    tomorrow = "\tsaturday";
```

In Zephir, strings don't support variable parsing like in PHP, you can use concatenation instead:

```
var name = "peter";
echo "hello: " . name;
```

Arrays

The array implementation in Zephir is basically the same as in PHP: Ordered maps optimized for several different uses; it can be treated as an array, list (vector), hash table (an implementation of a map), dictionary, collection, stack, queue, and probably more. As array values can be other arrays, trees and multidimensional arrays are also possible.

The syntax to define arrays is slightly different than in PHP:

```
//Square braces must be used to define arrays
let myArray = [1, 2, 3];

//Double colon must be used to define hashes' keys
let myHash = ["first": 1, "second": 2, "third": 3];
```

Only long and string values can be used as keys:

```
let myHash = [0: "first", 1: true, 2: null];
let myHash = ["first": 7.0, "second": "some string", "third": false];
```

Objects

Zephir allows to instantiate, manipulate, call methods, read class constants, etc from PHP objects:

```
let myObject = new stdClass(),
    myObject->someProperty = "my value";
```

1.7.2 Static Types

Static typing allows the developer to declare and use some variable types available in C. Variables can't change their type once they're declared as dynamic types. However, they allow the compiler to do a better optimization job. The following types are supported:

Type	Description
boolean	A boolean expresses a truth value. It can be either 'true' or 'false'.
integer	Signed integers. At least 16 bits in size.
unsigned integer	Unsigned integers. At least 16 bits in size.
char	Smallest addressable unit of the machine that can contain basic character set.
unsigned char	Same size as char, but guaranteed to be unsigned.
long	Long signed integer type. At least 32 bits in size.
unsigned long	Same as long, but unsigned.
float/double	Double precision floating-point type. The size is platform-dependent.
string	A string is series of characters, where a character is the same as a byte.

Boolean

A boolean expresses a truth value. It can be either 'true' or 'false'. Contrary to the dynamic behavior static boolean types remain boolean (true or false) no mater what value is assigned to them:

```
boolean a;
```

```
let a = true,
    a = 100, // automatically casted to true
    a = null, // automatically casted to false
    a = "hello"; // throws a compiler exception
```

Integer/Unsigned Integer

Integer values are like the integer member in dynamic values. Values assigned to integer variables remain integer:

```
int a;
let a = 50,
    a = -70,
    a = 100.25, // automatically casted to 100
    a = null, // automatically casted to 0
    a = false, // automatically casted to 0
    a = "hello"; // throws a compiler exception
```

Unsigned integers are like integers but they don't have sign, this means you can't store negative numbers in these sort of variables:

```
let a = 50,
    a = -70, // automatically casted to 70
    a = 100.25, // automatically casted to 100
    a = null, // automatically casted to 0
    a = false, // automatically casted to 0
    a = "hello"; // throws a compiler exception
```

Unsigned integers are twice bigger than standard integers, assign unsigned integers to integers may represent loss of data:

1.7. Types 17

```
uint a, int b;
let a = 2147483648,
    b = a, // possible loss of data
```

Long/Unsigned Long

Long variables are twice bigger than integer variables, thus they can store bigger numbers, As integers values assigned to long variables are automatically casted to this type:

```
long a;

let a = 50,
    a = -70,
    a = 100.25, // automatically casted to 100
    a = null, // automatically casted to 0
    a = false, // automatically casted to 0
    a = "hello"; // throws a compiler exception
```

Unsigned longs are like longs but they aren't signed, this means you can't store negative numbers in these sort of variables:

```
let a = 50,
    a = -70, // automatically casted to 70
    a = 100.25, // automatically casted to 100
    a = null, // automatically casted to 0
    a = false, // automatically casted to 0
    a = "hello"; // throws a compiler exception
```

Unsigned longs are twice bigger than standard longs, assign unsigned longs to longs may represent loss of data:

```
ulong a, long b;
let a = 4294967296,
    b = a, // possible loss of data
```

Char/Unsigned Char

Char variables are the smallest addressable unit of the machine that can contain basic character set. Every 'char' variable represents every character in a string:

```
char ch, string name = "peter";
let ch = name[2]; // stores 't'
```

1.8 Operators

Zephir's set of operators are similar to the ones PHP and also inherits some of their behaviors.

1.8.1 Arithmetic Operators

The following operators are supported:

Operation	Example
Negation	-a
Addition	a + b
Substraction	a - b
Multiplication	a * b
Division	a/b
Modulus	a % b

1.8.2 Comparison Operators

Comparison operators depend on the type of variables compared, for example, if both compared operands are dynamic variables the behavior is the same as in PHP:

a == b	Equal	TRUE if a is equal to b after type juggling.
a === b	Identical	TRUE if a is equal to b, and they are of the same type.
a != b	Not equal	TRUE if a is not equal to b after type juggling.
a <> b	Not equal	TRUE if a is not equal to b after type juggling.
a !== b	Not identical	TRUE if a is not equal to b, or they are not of the same type.
a < b	Less than	TRUE if a is strictly less than b.
a > b	Greater than	TRUE if a is strictly greater than b.
a <= b	Less than or equal to	TRUE if a is less than or equal to b.
a >= b	Greater than or equal to	TRUE if a is greater than or equal to b.

Learn more about comparison of dynamic variables in the php manual.

1.8.3 Special Operators

The following operators are supported:

Empty

This operator allows to check whether an expression is empty. 'Empty' means the expression is null, is an empty string or an empty array:

```
let someVar = "";
if empty someVar {
    echo "is empty!";
}

let someVar = "hello";
if !emptyVar someVar {
    echo "is not empty!";
}
```

Isset

This operator checks whether a property or index has been defined in an array or object:

```
let someArray = ["a": 1, "b": 2, "c": 3];
if isset someArray["b"] { // check if the array has an index "b"
    echo "yes, it has an index 'b'\n";
}
```

1.8. Operators 19

Using 'isset' as return expression:

```
return isset this->{someProperty};
```

Fetch

'Fetch' is an operator that reduce a common operation in PHP into a single instruction:

```
<?php
if (isset($myArray[$key])) {
    $value = $myArray[$key];
    echo $value;
}</pre>
```

In Zephir, you can write the same code as:

```
if fetch value, myArray[key] {
    echo value;
}
```

'Fetch' only returns true if the 'key' is a valid item in the array, only in that case, 'value' is populated.

Type Hints

Zephir always tries to check whether an object implements methods and properties called/accessed on a variable that is inferred to be an object:

```
let o = new MyObject();

// Zephir checks if "myMethod" is implemented on MyObject
o->myMethod();
```

However, due to the dynamism inherited from PHP, sometimes it is not easy to know the class of an object so Zephir can not produce errors reports effectively. A "type hint" tells the compiler which class is related to a dynamic variable allowing the compiler to perform more compilation checks:

```
// Tell the compiler that "o"
// is an instance of class MyClass
let o = <MyClass> this->_myObject;
o->myMethod();
```

1.9 Arrays

Array manipulation in Zephir provides a way to use PHP arrays. An array is an implementation of a hash table.

1.9.1 Creating Arrays

An array is created enclosing their elements in square brackets:

```
//Creating an empty array
let elements = [];
//Creating an array with elements
```

```
let elements = [1, 3, 4];

//Creating an array with elements of different types
let elements = ["first", 2, true];

//A multidimensional array
let elements = [[0, 1], [4, 5], [2, 3]];

As PHP, hashes or dictionaries are supported:

//Creating a hash with string keys
let elements = ["foo": "bar", "bar": "foo"];

//Creating a hash with numeric keys
let elements = [4: "bar", 8: "foo"];

//Creating a hash with mixed string and numeric keys
let elements = [4: "bar", "foo": 8];
```

1.9.2 Updating arrays

Arrays are updated in the same way as PHP using square brackets:

```
//Updating an array with a string key
let elements["foo"] = "bar";

//Updating an array with a numeric key
let elements[0] = "bar";
```

1.9.3 Appending elements

Elements can be appended at the end of the array as follows:

```
//Append an element to the array
let elements[] = "bar";
```

1.9.4 Reading elements from arrays

Is possible to read array elements as follows:

```
//Getting an element using the string key "foo"
let foo = elements["foo"];

//Getting an element using the numeric key 0
let foo = elements[0];
```

1.10 Classes and Objects

Zephir promotes object-oriented programming, this is why you can only export methods and classes in extensions, also you will see that most of the time, runtime errors raise exceptions instead of fatal errors or warnings.

1.10.1 Classes

Every Zephir file must implement a class or an interface (and just once). A class structure is very similar to a PHP class:

```
namespace Test;

/**
  * This is a sample class
  */
class MyClass
{
```

1.10.2 Implementing Methods

The "function" keyword introduces a method. Methods implements the usual visibility modifiers available in PHP, explicity set a visibility modifier is mandatory in Zephir:

Methods can receive required and optional parameters:

```
namespace Test;

class MyClass
{
    /**
    * All parameters are required
    */
    public function doSuml(a, b)
    {
        return a + b;
    }

    /**
    * Just 'a' is required, 'b' is optional and it has a default value
```

```
public function doSum2(a, b=3)
{
    return a + b;
}

/**
    * Both parameters are optional
    */
public function doSum3(a=1, b=2)
{
    return a + b;
}

/**
    * Parameters are required and their values must be integer
    */
public function doSum4(int a, int b)
{
    return a + b;
}

/**
    * Static typed with default values
    */
public function doSum4(int a=4, int b=2)
{
    return a + b;
}
```

Public Visibility

Methods marked as "public" are exported to the PHP extension, this means that public methods are visible to the PHP code as well to the extension itself.

Protected Visibility

Methods marked as "protected" are exported to the PHP extension, this means that protected methods are visible to the PHP code as well to the extension itself. However, protected methods can only be called in the scope of the class or in classes that inherit them.

Private Visibility

Methods marked as "private" are not exported to the PHP extension, this means that private methods are only visible to the class where they're implemented.

1.10.3 Implementing Properties

Class member variables are called "properties". By default, they act as PHP properties. Properties are exported to the PHP extension and are visibles from PHP code. Properties implement the usual visibility modifiers available in PHP, explicity set a visibility modifier is mandatory in Zephir:

```
namespace Test;

class MyClass
{
    public myProperty1;
    protected myProperty2;
    private myProperty3;
}
```

Within class methods non-static properties may be accessed by using -> (Object Operator): this->property (where property is the name of the property):

```
namespace Test;

class MyClass
{

    protected myProperty;

    public function setMyProperty(var myProperty)
    {
        let this->myProperty = myProperty;
    }

    public function getMyProperty()
    {
        return this->myProperty;
    }
}
```

Properties can have literal compatible default values. These values must be able to be evaluated at compile time and must not depend on run-time information in order to be evaluated:

```
namespace Test;

class MyClass
{

    protected myProperty1 = null;
    protected myProperty2 = false;
    protected myProperty3 = 2.0;
    protected myProperty4 = 5;
    protected myProperty5 = "my value";
}
```

1.10.4 Updating Properties

Properties can be updated by accessing them using the '->' operator:

```
let this->myProperty = 100;
```

Zephir checks that properties do exist when a program is accessing them, if a property is not declared you will get a compiler exception:

```
CompilerException: Property '_optionsx' is not defined on class '_optionsx' in /Users/scott/cphalcon this->_optionsx = options;
```

If you want to avoid this compiler validation or just create a property dynamically, you can enclose the property name using string quotes:

```
let this->{"myProperty"} = 100;
```

You can also use a simple variable to update a property, the property name will be taken from the variable:

```
let someProperty = "myProperty";
let this->{someProperty} = 100;
```

1.10.5 Reading Properties

Properties can be read by accessing them using the '->' operator:

```
echo this->myProperty;
```

As when updating, properties can be dynamically read this way:

```
//Avoid compiler check or read a dynamic user defined property
echo this->{"myProperty"};

//Read using a variable name
let someProperty = "myProperty";
echo this->{someProperty}
```

1.10.6 Class Constants

Class may contain class constants that remain the same and unchangeable once the extension is compiled. Class constants are exported to the PHP extension allowing them to be used from PHP.

```
namespace Test;

class MyClass
{
    const MYCONSTANT1 = false;
    const MYCONSTANT2 = 1.0;
}
```

Class constants can be accessed using the class name and the static operator (::):

```
namespace Test;

class MyClass
{
    const MYCONSTANT1 = false;
    const MYCONSTANT2 = 1.0;
```

```
public function someMethod()
        return MyClass::MYCONSTANT1;
1.10.7 Calling Methods
Methods can be called using the object operator (->) as in PHP:
namespace Test;
class MyClass
    protected function _someHiddenMethod(a, b)
        return a - b;
    public function someMethod(c, d)
        return this->_someHiddenMethod(c, d);
}
Static methods must be called using the static operator (::):
namespace Test;
class MyClass
    protected static function _someHiddenMethod(a, b)
        return a - b;
    public static function someMethod(c, d)
        return self::_someHiddenMethod(c, d);
```

You can call methods in a dynamic manner as follows:

```
namespace Test;
class MyClass
    protected adapter;
    public function setAdapter(var adapter)
```

```
let this->adapter = adapter;
}

public function someMethod(var methodName)
{
    return this->adapter->{methodName}();
}
```

1.10.8 Getter/Setter shortcuts

Like in C#, you can use get/set/toString shortcuts in Zephir, this feature allows to easily write setters and getters for properties without explictly implementing those methods as such.

For example, without shortcuts we could find code like:

```
namespace Test;

class MyClass
{
    protected myProperty;

    protected someProperty = 10;

    public function setMyProperty(myProperty)
    {
        this->myProperty = myProperty;
    }

    public function getMyProperty()
    {
        return this->myProperty;
    }

    public function setSomeProperty(someProperty)
    {
        this->someProperty = someProperty;
    }

    public function getSomeProperty()
    {
        return this->someProperty;
    }

    public function __toString()
    {
        return this->myProperty;
    }
}
```

You can write the same code using shortcuts as follows:

```
namespace App;
class MyClass
{
```

```
protected myProperty {
    set, get, toString
};

protected someProperty = 10 {
    set, get
}
```

When the code is compiled those methods are exported as real methods but you don't have to write them one by one.

1.10.9 Return Type Hints

Mthods in classes and interfaces can have return type hints, these will provide useful extra information to the compiler to inform you about errors in your application. Consider the following example:

```
namespace App;
class MyClass
   public function getSomeData() -> string
        // this will throw a compiler exception
        // since the returned value (boolean) does not match
        // the expected returned type string
        return false;
   public function getSomeOther() -> <App\MyInterface>
        // this will throw a compiler exception
        // if the returned object does not implement
        // the expected interface App\MyInterface
        return new App\MyObject;
   public function process()
    {
       var myObject;
        // the type-hint will tell the compiler that
        // myObject is an instance of a class
        // that implement App\MyInterface
        let myObject = this->getSomeOther();
        // the compiler will check if App\MyInterface
        // implements a method called "someMethod"
        echo myObject->someMethod();
    }
```

1.11 Control Structures

Zephir implements a simplified set of control structures present in similar languages like C, PHP etc.

1.11.1 Conditionals

If Statement

'if' statements evaluates an expression executing this trace if the evaluation is true. Braces are compulsory, an 'if' can have an optional 'else' clause, multiple 'if'/'else' constructs can be chained together:

```
if false {
    echo "false?";
} else {
    if true {
       echo "true!";
    } else {
       echo "neither true nor false";
    }
}
```

Parentheses in the evaluated expression are optional:

```
if a < 0 { return -1; } else { if a > 0 { return 1; } }
```

Switch Statement

A 'switch' evalutes an expression against a series of predefined literal values executing the corresponding 'case' block or falling back to the 'default' block case:

```
switch count(items) {
    case 1:
    case 3:
        echo "odd items";
        break;
    case 2:
    case 4:
        echo "even items";
        break;
    default:
        echo "unknown items";
}
```

1.11.2 Loops

While Statement

'while' denotes a loop that iterates as long as its given condition evaluates as true:

```
let counter = 5;
while counter {
    let counter -= 1;
}
```

Loop Statement

In addition to 'while', 'loop' can be used to create infinite loops:

```
let n = 40;
loop {
    let n -= 2;
    if n % 5 == 0 { break; }
    echo x, "\n";
}
```

For Statement

A 'for' is a control structure that allows to traverse arrays or strings:

```
for item in ["a", "b", "c", "d"] {
    echo item, "\n";
}
```

Keys in hashes can be obtained in the following way:

```
let items = ["a": 1, "b": 2, "c": 3, "d": 4];
for key, value in items {
    echo key, " ", value, "\n";
}
```

A 'for' loop can also be instructed to traverse an array or string in reverse order:

```
let items = [1, 2, 3, 4, 5];
for value in reverse items {
    echo value, "\n";
}
```

A 'for' can be used to traverse string variables:

```
string language = "zephir"; char ch;
for ch in language {
    echo "[", ch "]";
}
```

In reverse order:

```
string language = "zephir"; char ch;
for ch in reverse language {
    echo "[", ch "]";
}
```

A standard 'for' that traverses a range of integer values can be written as follows:

```
for i in range(1, 10) {
    echo i, "\n";
}
```

Break Statement

'break' ends execution of the current 'while', 'for' or 'loop' statements:

```
for item in ["a", "b", "c", "d"] {
    if item == "c" {
        break; // exit the for
    }
    echo item, "\n";
}
```

Continue Statement

'continue' is used within looping structures to skip the rest of the current loop iteration and continue execution at the condition evaluation and then the beginning of the next iteration.

```
let a = 5;
while a > 0 {
    let a--;
    if a == 3 {
        continue;
    }
    echo a, "\n";
}
```

1.11.3 Require

The 'require' statement dynamically includes and evaluates a specified PHP file. Note that files included via Zephir are interpreted by Zend Engine as normal PHP files. 'require' does not allow to include other zephir files in runtime.

```
if file_exists(path) {
    require path;
}
```

1.11.4 Let

'Let' statement is used to mutate variables, properties and arrays. Variables are by default inmutable and this instruction makes them mutable:

Also this instruction must be used to increment/decrement variables:

Multiple mutations can be performed in a single 'let' operation:

```
let price = 1.00, realPrice = price, status = false;
```

1.12 Calling Functions

PHP has a rich library of functions you can use in your extensions. To call a PHP function, you can just refer its name in the Zephir code.

```
namespace MyLibrary;

class Encoder
{
    public function encode(var text)
    {
        if strlen(text) != 0 {
            return base64_encode(text);
        }
        return false;
    }
}
```

You can call also functions that are expected to exist in the PHP userland but they aren't built-in with PHP:

```
namespace MyLibrary;

class Encoder
{
    public function encode(var text)
    {
        if strlen(text) != 0 {
            if function_exists("my_custom_encoder") {
                return my_custom_encoder($text);
        } else {
                return base64_encode(text);
        }
        return false;
    }
}
```

Note that all PHP functions only receive and return dynamic variables, if you pass a static typed variable as a parameter, some temporary dynamic variable will be used as a bridge in order to call them:

}

Similarly, functions return dynamic values that cannot be directly assigned to static variables without the appropriate cast:

```
namespace MyLibrary;

class Encoder
{

   public function encode(string text)
   {
      string encoded = "";

      if strlen(text) != 0 {
            let encoded = (string) base64_encode(text);
            return '(' . encoded . ')';
      }
      return false;
   }
}
```

Sometimes, we would need to call functions in a dynamic way, you can call them as follows:

```
namespace MyLibrary;

class Encoder
{
    public function encode(var callback, string text)
    {
        return {callback}(text);
    }
}
```

1.13 Using optimizers

Most common functions in Zephir use internal optimizers. An 'optimizer' works like an interceptor for function calls. An 'optimizer' replaces the call for the function in the PHP userland by direct C-calls which are faster and have a lower overhead improving performance.

To create an optimizer you have to create a class in the 'optimizers' directory, the following convention must be used:

Function in Zephir	Optimizer Class Name	Optimizer Path	Function in C
calculate_pi	CalculatePiOptimizer	optimizers/CalculatePiOptimizer.php	my_calculate_pi

This is the basic structure for an 'optimizer':

```
}
```

Implementation of optimizers highly depends on the kind of code you want to generate. In our example, we're going to replace the call to this function by a call to a c-function. In Zephir, the code used to call this function is:

```
let pi = calculate_pi(1000);
```

So, the optimizer will expect just one parameter, we have to validate that to avoid problems later:

public function optimize(array \$expression, Call \$call, CompilationContext \$context)
{

 if (!isset(\$expression['parameters'])) {
 throw new CompilerException("'calculate_pi' requires one parameter");
 }

 if (count(\$expression['parameters']) < 2) {
 throw new CompilerException("'calculate_pi' requires one parameter");
 }

 //...
}
</pre>

There are functions that are just called and they don't return any value, our function returns a value that is the calculated PI value. So we need to be aware that the type of the variable used to received this calculated value is OK:

```
public function optimize(array $expression, Call $call, CompilationContext $context)
{
    if (!isset($expression['parameters'])) {
        throw new CompilerException("'calculate_pi' requires one parameter");
    }
    if (count($expression['parameters']) < 2) {
        throw new CompilerException("'calculate_pi' requires one parameter");
    }
    /**
    * Process the expected symbol to be returned
    */
    $call->processExpectedReturn($context);
    $symbolVariable = $call->getSymbolVariable();
    if ($symbolVariable->isNotDouble()) {
        throw new CompilerException("Calculated PI values only can be stored in double variables", $context of the context of th
```

We're checking if the value returned will be stored in a variable type 'double', if not a compiler exception is thrown.

The next thing we need to do is process the parameters passed to the function:

```
<?php
$resolvedParams = $call->getReadOnlyResolvedParams($expression['parameters'], $context, $expression)
```

As a good practice with Zephir is important to create functions that don't modify their parameters, if you are changing the parameters passed, Zephir will need to allocate memory for constants passed and you have to use getResolved-Params instead of getReadOnlyResolvedParams.

Code returned by these methods is valid C-code that can be used in the code printer to generate the c-function call:

```
<?php
//Generate the C-code
return new CompiledExpression('double', 'calculate_pi(' . $resolvedParams[0] . ')', $expression);</pre>
```

All optimizers must return a CompiledExpression instance, this will tell the compiler the type returned by the code and its related C-code.

The complete optimizer code is:

```
<?php
class CalculatePiOptimizer extends OptimizerAbstract
   public function optimize(array $expression, Call $call, CompilationContext $context)
        if (!isset($expression['parameters'])) {
            throw new CompilerException("'calculate_pi' requires one parameter");
        if (count($expression['parameters']) < 2) {</pre>
            throw new CompilerException("'calculate_pi' requires one parameter");
        }
         * Process the expected symbol to be returned
        $call->processExpectedReturn($context);
        $symbolVariable = $call->getSymbolVariable();
        if ($symbolVariable->isNotDouble()) {
            throw new CompilerException ("Calculated PI values only can be stored in double variables
        }
        $resolvedParams = $call->getReadOnlyResolvedParams($expression['parameters'], $context, $exp.
        return new CompiledExpression('double', 'my_calculate_pi(' . $resolvedParams[0] . ')', $exp:
    }
```

1.14 License

}

Copyright (c) 2013-2014 Zephir Team and contributors http://zephir-lang.com

1.14. License 35

MIT License

Permission is hereby granted, free of charge, to any person obtaining a copy of this software and associated documentation files (the "Software"), to deal in the Software without restriction, including without limitation the rights to use, copy, modify, merge, publish, distribute, sublicense, and/or sell copies of the Software, and to permit persons to whom the Software is furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE SOFTWARE.

Other Formats

• PDF

Index

l index index, 37